Лабораторная работа № 2

Файлы

1. Цель работы

В языке Паскаль существует три вида файлов: текстовые, типизированные, нетипизированные (бестиповые).

Общая цель работы - систематизация сведений о различных видах файлов и операциях с ними и сравнение перечисленных видов файлов с  точки зрения их использования. 

Частная цель - освоении средств обработки типизированных и нетипизированных файлов и дополнительных средств обработки текстовых файлов в языке Паскаль.

Общие представления о файлах и навыки работы с текстовыми файлами получены в первой части курса, однако для формирования целостной картины ниже воспроизведены некоторые из уже известных положений.

2. Аспекты представления файла 

При использовании файлов в программе (в качестве источников входных данных или для записи выходных данных) файл представляется в двух аспектах:

- физическом; это собственно файл с данными на внешнем носителе (диске); идентифицируется именем согласно правилам операционной системы (MS DOS, Windows и т.д.;

- логическом; это представление файла в программе, зависящее от правил языка программирования; в Паскале файл в программе представляется файловой переменной, имя которой строится по обычным правилам Паскаля.

Входной файл создается любым подходящим способом независимо от программы и до ее запуска. Выходной файл формируется программой.

Соответствие между физической и логической организацией устанавливается специальным оператором языка, связывающим физическое имя файла с файловой переменной; с этого момента программа имеет дело только с файловой переменной, и термин «файл» применительно к операторам обработки файлов относится к файловой переменной (корректная формулировка звучала бы, например, так: «... чтение из файла, представленного файловой переменной f» вместо краткого: «чтение из файла f »).

3. Виды файлов и их описание
3.1. Текстовый файл

( Определение и структура
Текстовый файл – совокупность строк (последовательностей символов) переменной длины, заканчивающихся специальным символом eoln (конец строки):

  <строка>             <строка>           <строка>      <строка>    


Первоначально любой файл данных создается как текстовый. Набранные на клавиатуре данные представляют собой  стандартный входной файл. Содержимое дисплея при просмотре любого файла – стандартный выходной файл. Эти файлы используются при задании и просмотре данных. Для хранения данных последние записываются в файл на внешнем запоминающем устройстве (диске).

( Описание файловой переменной


Var                                      Здесь:    

     <ф.п.>:text;                    ф.п. – файловая переменная, 

                                             text -  стандартный тип.

Стандартные файловые переменные – input (по умолчанию связан с клавиатурой) и    output (по умолчанию связан с дисплеем).

( Специфика


Компоненты-строки представлены во внешнем (символьном) виде: каждый байт являет собой код согласно таблице кодов.

В программе из текстовых файлов  и выводятся только неструктурированные данные –  числа, символы, строки. Структурированные данные (массивы, записи) необходимо вводить по компонентам структуры (элементам, полям).

Числа разделяются пробелами и/или символами eoln.

Символы считываются подряд (eoln – тоже символ!).

Строки считываются согласно соответствующим правилам, кратко упомянутым        в п. 2 лабораторной работы № 1.

( Преимущества текстовых файлов:

- простота создания (непосредственный набор данных на клавиатуре);

- наглядность (непосредственный просмотр средствами текстовых редакторов);

- универсальность (можно обойтись только этим видом файлов). 

3.2. Типизированный файл

( Определение и структура
Все компоненты такщго файла принадлежат к одному типу; тип может быть любым, кроме файлового:

   <компонент> <компонент> <компонент>               <компонент>


Компоненты размещены непосредственно один за другим.

Файл хранится только на внешнем запоминающем устройстве (в частности, на диске).

( Общая форма описания ( с определением файлового типа)
{Описание файлового типа}

             type

                 T=file of <тип компонентов>;

{Описание файловой переменной типа T }

             var

                <ф.п.>:T;

( Сокращенное описание


{Описание файловой переменной типа T }

             var

                <ф.п.>:file of <тип компонентов>;
Значение файловой переменной - файл типа T.
( Специфика

Информация в типизированных файлах хранится во внутреннем представлении.

Поэтому такой файл  может быть создан только программным путем  - путем записи содержимого некоторой переменной того же типа,  что и тип файла, из оперативной памяти (т.е. во внутреннем представлении) в файл:

       
                                           Программа       

Текстовый    чтение         создания        запись         Типизиров. 

файл                                   типизиров.                         файл 

                                           файла

По этой же причине такой файл нельзя просмотреть непосредственно,  как текст (при такой попытке содержимое файла, физически являющее собой  последовательность байтов, на экране будет представлено последовательностью символов, закодированных этими байтами согласно таблице кодов).  Для просмотра типизированный файл необходимо перевести также программным путем в соответствующий текстовый (на экране или диске) файл: 


                                           Программа       

Типизиров.    чтение        создания        запись         Текстовый 

файл                                   текстового                         файл 

                                           файла

Компоненты типизированного файла считываются и записываются целиком.

( Преимущества типизированных файлов:

           - компактное хранение данных, для внешнего представления которых требуется значительно больше памяти, чем для внутреннего (числа, записи);

· удобство работы с компонентами сложной структуры (за счет чтения-записи компонентов целиком.


3.3. Нетипизированный файл (бестиповый)

( Определение и структура

Файл рассматривается как последовательность байтов, заканчивающаяся символом eof, и обрабатывается  (считывается и записывается) блоками записей  - совокупностей байтов заданной программистом длины. 

Размещается только на внешнем запоминающем устройстве (в частности, на диске).
( Описание


var
    <ф.п.>:file;    
( Специфика

Нетипизированный файл совместим с файлом любого типа. Это означает, что любой уже созданный файл (текстовый, типизированный, нетипизированный) может быть открыт и считан  как нетипизированный, по байтам; с другой стороны, любое данное из оперативной памяти может быть по байтам записано в нетипизированный файл, который затем можно открыть и обрабатывать как текстовый, типизированный или нетипизирован-

ный. 

Обработка файла как нетипизированного позволяет организовать высокоскоростной обмен между диском и памятью. Обмен осуществляется записями – последовательностями байтов, длину которых задает программист в зависимости от смысла (интерпретации), вкладываемого в эту последовательность при дальнейшей обработке.

4. Операции над файлами

Процедуры и функции для работы с файлами представлены заголовками. Обязательный параметр всех операций – файловая переменная (ф.п.). Этот параметр обладает двумя особенностями: 

( всегда передается как параметр-переменная (var);

( описывается как нетипизированный параметр,  так как соответствующие 
фактические параметры могут принадлежать к различным файловым типам.

4.1. Общие операции

( Установочные и завершающие

procedure assign(var <ф.п.>;<имя файла>:string); - назначение имени внешнего файла файловой переменной (связывание имени файла с файловой переменной);


procedure  reset (var  <ф.п.> [:file;<длина записи>:word]  ); - открытие существующего файла (файла, связанного с файловой переменной <ф.п.>) для чтения;

procedure rewrite(var <ф.п.>[:file;<длина записи>:word] ); - создание и открытие нового файла для записи;

procedure close(var  <ф.п.>); - закрытие открытого файла.

( Ввод-вывод

procedure read(var  <ф.п.>;<список ввода>); - чтение данных из текстового или типизированного файла согласно списку ввода; элемент списка ввода для текстового файла – число или символ или строка string, для типизированного – компонент файла; 

procedure write(var  <ф.п.>;<список вывода>); - запись данных в текстовый или типизированный файл согласно списку вывода; элемент списка вывода для текстового файла – число или символ или строка string, для типизированного – компонент файла; 

 procedure blockread(var  <ф.п.>:file; ... ); - чтение         для нетипизированного

procedure blockwrite(var  <ф.п.>:file; ... ); -  запись        файла; см. п. 4.4.

( Проверка конца файла

function eof(var var  <ф.п.>):boolean; - тестирование конца файла; возвращает true, если файловый указатель стоит в конце файла. При записи это означает, что очередной компонент будет добавлен в конец файла, при чтении – что файл исчерпан.

4.2. Операции над текстовыми файлами

procedure append (var <ф.п.>:text); - открытие файла для расширения; указатель файла устанавливается в конец; если файл был открыт, то он закрывается и открывается вновь, но уже для добавления;

procedure readln(var  <ф.п.>:text;<список ввода>); - чтение данных согласно списку ввода и переход на следующую строку; 

procedure writeln(var  <ф.п.>:text;<список вывода>); - запись данных в файл согласно списку вывода с добавлением в конце выведенной строки маркера конца строки (переход на следующую строку). 

4.3. Операции над типизированными файлами
 procedure seek (var <ф.п.>; <№ компонента>:longint); - обращение к компоненту с заданным номером (прямой доступ); первый компонент имеет № 0!

function filepos(var <ф.п.>):longint; - возвращает текущее положение файлового указателя (порядковый номер подлежащего обработке компонента); 

function filesize(var <ф.п.>):longint; - возвращает размер файла (количество компонентов).

Для типизированных файлов возможна запись в файл, открытый для чтения.

4.4.  Операции над нетипизированными файлами

( Запись, блок, буфер

Как уже упоминалось, нетипизированный файл для программы – всего лишь последовательность байтов. 

Запись – минимальная единица, которую можно считать или записать в файл. Для программы это последовательность байтов заданной длины.

Длина записи определяется программистом исходя из того, какую группу данных он хотел бы считывать или записывать не более чем за одно обращение к внешнему устройству (одно целое число; массив чисел; массив строк; запись record и т.д.). Для оптимизации временных характеристик программы лучше брать длину записи кратной 512 (длине кластера).

Указывается длина записи при открытии файла (см. ниже).

Блок – совокупность  заданного числа записей, считываемых или записываемых за одно обращение к диску. Таким образом, блок может состоять из одной или более записей. Число записей в блоке задается в процедурах ввода-вывода.

Буфер – переменная, участвующая в обмене с диском. Значение именно этой переменной считывается или записывается в файл, т.е. ее содержимое – это упомянутый блок записей. Имя этой переменной указывается в процедурах ввода-вывода.
Общая структура нетипизированного файла и схема обработки


 



                                                чтение-     

                                                 запись


Важно следующее:

( запись реально должна представлять собой осмысленную группу байтов;

( длина записи тогда будет равна длине этой группы;

( пусть решено осуществлять обмен блоками по k записей; тогда длина буфера должна быть равна k* <длина записи>, т.е. в качестве буфера должна фигурировать переменная такой длины.

Например, желательно записать в нетипизированный файл последовательность чисел типа integer.
Пусть запись содержит одно число; тогда ее длина равна 2 байта.

Пусть мы хотим писать в файл блоки по 10 записей. Тогда длина буфера должна быть равна 2*10=20 байтов. Объявим переменную-массив из 10 элементов типа integer; этот массив и будем использовать в качестве буфера.

Полный пример приведен в п. 5. 

( Открытие и закрытие файлов

procedure  reset (var  <ф.п.> :file;<длина записи>:word ); - открытие существующего файла (файла, связанного с файловой переменной <ф.п.>) для чтения;

procedure rewrite(var <ф.п.>:file;<длина записи>:word ); - создание и открытие нового файла для записи;

( Ввод - вывод
procedure blockread(var  <ф.п.>:file;  var <буфер>; <число записей>:word [;<результат>:word]); - чтение из файла в переменную <буфер> блока из  <число записей>*<длина записи> байтов; необязательный параметр <результат> - число фактически считанных записей (для случая, когда в последнем блоке записей  меньше заданного числа);

procedure blockwrite(var  <ф.п.>:file;  var <буфер>; <число записей>:word [;<результат>:word]); - запись в  файл значения переменной <буфер>; смысл остальных параметров тот же, что и для чтения.

( Можно использовать процедуры, определенные для титпизированных файлов:

seek, filepose, filesize.
5. Пример


Пусть в файл txt.txt, созданный обычным текстовым редактором, записана информация о книгах. Описание каждой книги имеет вид:



                                                                                       refdat    

                        title                                       year                           

                               
                             30 символов                                 1..2000 


Пусть во входном файле каждое поле ссылки размещается в отдельной строке, т.е. входная форма имеет вид:

            


<название>                 для первой

<год издания>            ссылки 

    ...........

<название>                 для последней

<год издания>            ссылки

Задание 

1. Добавить в файл одну новую ссылку, считав ее с клавиатуры.

2. Рассматривая файл из п. 1 как нетипизированный, переписать его в другой нетипизированный файл блоками, содержащими одну запись длиной в 1 байт. Просмотрев оба файла, убедится, что они идентичны.

3. Создать на основе исходного файла типизированный.

4. Рассматривая файл из п. 3 как нетипизированный, переписать его в другой нетипизированный  файл блоками  длиной, равной длине записи-ссылки (record). Получим два идентичных файла во внутреннем представлении.

5. Открыв файл из п.3 и файл, полученный в п.4, как типизированные, переписать их в текстовые. Убедиться в идентичности полученных файлов.

6. В типизированный файл из п. 3 добавить одну запись; поменять местами 3-ю и 5-ю записи. Просмотреть результаты.

Пояснения к заданию

Каждый из упоминаемых в задании файлов, кроме исходного, является результатом выполнения некоторого пункта (где файл открывается для записи) и входом следующего пункта (где он открывается для чтения). Будем использовать различные файловые переменные для открытия файла как входного и выходного.


Подход к файлу как к нетипизированному – по сути дела абстрагирование от смысла записанной в нем информации. Поэтому один и тот же физический файл может быть открыт двумя способами – в соответствии со своим типом и как нетипизированный.


Отобразим эти моменты в именах файловых переменных, конструируя эти имена из двух частей:



<тип файла>_<способ открытия> - для входного файла,

<тип файла-источника>_<тип создаваемого файла> - для выходного файла.

Закодируем тип файла так:

txt - текстовый,

typ - типизированный,

ntp – нетипизированный. 

Способ открытия файла можно обозначить так же.

Например, если файл создается в некотором пункте задания и файловая переменная имеет имя txt_typ, то речь идет о типизированном файле, создаваемом из текстового; для входного файла такого имени быть не может, так как текстовый файл может быть открыт либо как текстовый (и тогда мы назовем файловую переменную txt_txt), либо как нетипизированный (txt.ntp).

Для наглядности представим задание графической схемой с элементами:




                                          <тип файла>        

                                         <физич. имя>

 
Имена физических файлов сформируем, заменив “_” на “(” в имени файловой переменной, с которой файл создавался.



                      Текстовый                                              Текстовый
                        txt.txt                                                           txt.txt





      Нетипизир.                                 Типизир.                                                      Типизир.

          ntp.ntp                                        txt.typ                                                            txt.typ




                                  Нетипизир.                                              Текстовый
                                typ.ntp                                               typ.txt


                  


                 Текстовый
                    ntp.txt

Соответствующий алгоритм преобразований тривиален и требует только правильного использования конструкций языка программирования. 

Для обработки записи-ссылки, организации обмена блоками и изменения типизированного файла потребуются переменные. Пусть это следующие переменные:

 r  – обрабатываемая запись-ссылка; 

buf_rec – буфер для копирования записями (record)       тип refdat
при блоковом вводе-выводе;

recs – длина записи-ссылки (record); она же – размер буфера; тип integer;

buf_byte  ( буфер для копирования по байтам при блоковом вводе-выводе; тип byte. 

 Далее приведем непосредственно программу, сопроводив ее комментариями.

  program  transform_of_files(txt_txt, txt_ntp, ntp_ntp, txt_typ, typ_typ, typ_ntp, typ_ntp1, typ_txt,

                                                                                                                                   ntp_typ, ntp_txt);                                                                                                                                                                  

      type {тип ссылки на книгу}

           retdat = record

                             title: string [30] 

                             year: 1..2000;

                         end;  

      var

           txt_txt, typ_txt, ntp_txt: text;                    {текстовые файлы                                       }

           txt_typ, typ_typ, ntp_typ: file of  retdat ;  {типизированные файлы                           }

           txt_ntp, ntp_ntp, typ_ntp, typ_ntp1: file;  {нетипизированные файлы                       }

           r, buf_rec: retdat;   {запись-ссылка и буфер для копирования записями (record) }

           recs: integer;          {длина записи-ссылки (record); она же – размер буфера         }

           buf_byte: byte;       { буфер для копирования по байтам                                           }

      begin
            {----- 1. Добавить в файл одну новую ссылку, считав ее с клавиатуры -------------}                                          

             assign (txt_txt, ‘txt.txt’);              {назначение исходного файла                 }
             append (txt_txt);                          {открытие его для добавления                }

             readln (r. title); readln (r. year); {чтение по полям записи с клавиатуры } 
             write (txt_txt, r);                          {запись в конец файла                              }
             close (txt_txt);                             {закрытие измененного файла                 }

{----- 2. Рассматривая файл как нетипизированный, переписать его в другой ---}                                                                                                                        {-------  нетипизированный блоками по 1 байту. -------------------------------------------} 

             assign (txt_ntp, ‘txt.txt’);   {назначение входного файла                                                  }

             reset (txt_ntp, 1);      {открытие как нетипизир. для чтения с длиной записи в 1 байт}
             assign (ntp_ntp, ‘ntp.ntp’) { назначение выходного файла}

             rewrite (ntp_ntp, 1); {открытие как нетипизир. для записи с длиной записи в 1 байт }
             while not eof (txt_ntp) do {копирование блоками в 1 запись длиной 1 байт                  }
                   begin
                        blockread   (txt_ntp, buf_byte, 1); {чтение в   буфер buf_byte блока из1 записи  }          

                        blockwrite (ntp_ntp, buf_byte, 1); {запись из буфера buf_byte блока из 1 записи}

                   end;

             close (txt_ntp); close (ntp_ntp);                {закрытие файлов                                              }

            {Получили два идентичных файла txt.txt и ntp.ntp во внешнем представлении            }
            {----- 3. Создать на основе исходного файла типизированный --------------------------}

             assign (txt_txt, ‘txt.txt’);   {назначение входного файла                         }
             reset (txt_txt);                   {открытие как текстового для чтения           }
             assign (txt_typ, ‘txt.typ’); {назначение выходного файла                       }

             rewrite (txt_typ);              {открытие как типизированного для записи}

             while not eof (txt_txt
) do

                    begin

                         readln(txt_txt,r.title); readln(txt_txt,r.year);{чтение ссылки по полям}

                         write(txt_typ,r);       {запись компонента в типизир.файл целиком}

                    end;
             close (txt_txt); close (txt_typ);                {закрытие файлов                              }

 {----- 4. Рассматривая файл из п. 3 как нетипизированный, --------------------------}
 {переписать его в другой нетипизированный  файл блоками  длиной, равной  } 

 {длине записи-ссылки (record)                                                                                       }

             recs: = sizeof (retdat);        {определение длины записи-ссылки (размера типа)          }
             assign (typ_ntp, ‘txt.typ’);   {назначение входного файла                                                } 
 reset (typ_ntp, recs);        {открытие как нетипизир. для чтения с длиной записи recs} 

 assign (typ_ntp1, ‘typ.ntp’); {назначение выходного файла                                             }

 rewrite (typ_ntp1, recs);     {открытие как нетипизир. для записи с длиной записи recs}
             while not eof (typ_ntp) do  {копирование блоками в 1 запись длиной recs                      }
                   begin
                        blockread   (typ_ntp, buf_rec, 1);  {чтение в   буфер   buf_rec 1 записи                 }          

                        blockwrite (typ_ntp1, buf_rec, 1); {запись из буфера buf_rec 1 записи                  }

                   end;

             close (typ_ntp); close (typ_ntp1);                {закрытие файлов                                           }

             {Получили два идентичных файла txt.typ и typ.ntp во внутреннем представлении     }
 {----- 5. Открыв файл из п.3 и файл, полученный в п.4, как типизированные,     }

 {-------- переписать их в текстовые -------------------------------------------------------------}

 {Обработка файла из п. 3}
             assign (typ_typ, ‘txt.typ’); {назначение входного файла  из п. 3                                       }
             reset (typ_typ);                  {открытие как типизированного для чтения}

 assign (typ_txt, ‘typ.txt’);  { назначение выходного файла для переписи файла из п. 3}

 rewrite (typ_txt);               {открытие как текстового для записи                                     }

             while not eof (typ_typ) do

                    begin

                         read (typ_typ,r);                                       {чтение компонента-ссылки целиком }

                         writeln (typ_txt,r.title); writeln (typ_txt,r.year); {запись в текст. файл по полям }

                    end;
             close (typ_typ); close (typ_txt);                {закрытие файлов                                             }

             {Обработка файла из п. 4} 

             assign (ntp_typ, ‘typ.ntp’); {назначение входного файла  из п.4                                     } 

             reset (ntp_typ);                   {открытие как типизированного для чтения                        }

 assign (ntp_txt, ‘ntp.txt’);  {назначение выходного файла для переписи файла из п. 4}

 rewrite (ntp_txt);                {открытие как текстового для записи                                   }
             while not eof (ntp_typ) do

                    begin

                         read (ntp_typ,r);                                       {чтение компонента-ссылки целиком }

                         writeln (ntp_txt,r.title); writeln (ntp_txt,r.year); {запись в текст. файл по полям }

                    end;
             close (ntp_typ); close (ntp_txt);                {закрытие файлов} 

             {Получили два идентичных файла ntp.txt и typ.txt во внешнем представлении }
 {----- 6. В типизированный файл из п. 3 добавить одну запись; -----------------------}        

 {-------- поменять местами 3-ю и 5-ю записи. Просмотреть результаты ------------}
             assign (typ_typ, ‘txt.typ’);         {назначение входного файла  из п. 3                         }
             reset (typ_typ);      {открытие как типизированного; возможно и чтение, и запись  }

             readln (r. title); readln (r. year); {чтение ссылки с клавиатуры по полям                  }

             write (typ_typ, r);                       {добавление ссылки-нового компонента в конец    }
             seek (typ_typ, 2);          {установка  указателя файла на 3-ю запись                            }
             read (typ_typ, r);                  {читаем в r 3-ю запись                                                      }
             seek (typ_typ, 4);         {установка  указателя файла на 5-ю запись                            }                  

             read (typ_typ, buf _rec);      {читаем в buf _rec 5-ю запись                                          }

             seek (typ_typ, 2);           {установка  указателя файла на 3-ю запись                          }
             write (typ_typ,buf_rec);       {пишем buf_ rec в  3-ю запись                                         }
             seek (typ_typ, 4);          {установка  указателя файла на 5-ю запись                          }                  

             write (typ_typ,r);                  { пишем  r в  5-ю запись                                                  }
             <перевести ‘txt.typ’ в текстовый файл и просмотреть >
             <закрыть все открытые файлы>
       end.

6. Задание


Используя в качестве исходного файл с описаниями объектов из лабораторной работы № 1, выполнить задание, сформулированное в разделе “Пример”.


При необходимости задание может быть модифицировано по усмотрению преподавателя.

7. Контрольные вопросы

1. Что произойдет, если файл создан как типизированный, а читается как текстовый? Наоборот?

Есть ли ситуации, когда данные будут прочитаны верно?

2. Что общего в обработке типизированных и нетипизированных файлов? 

3. Всегда ли можно текстовый или типизированный файл открыть и прочитать как нетипизированный?

4. Можно ли нетипизированный файл открыть и правильно прочитать как текстовый? Как типизированный? В каких случаях?

5. Имеется некоторая процедура с параметром-файловой переменной:

           ...... ( ......; f:file; .....);
Верно ли записан параметр?

6. Вводятся последовательности значений троек переменных a,b,c следующих типов:

a: byte;

b: real;

c: record

       sym: char;

       num: byte;

    end;

Что и как надо сделать, чтобы сохранить эти значения в текстовом файле? Типизированном? Нетипизированном?

7. Что следовало бы сделать, если бы в п. 6 задания из примера файл был открыт не для записи, а для чтения?

Перевезенцева Е.С., Шамаева О.Ю.

ОБРАБОТКА ДАННЫХ СЛОЖНОЙ СТРУКТУРЫ В ЯЗЫКЕ ПАСКАЛЬ.

Лабораторные работы по курсу «Алгоритмические языки и программирование».

 
/Под ред. В.П. Климанова -М.: Изд-во МЭИ,1999. - 48 с.

